数学实验讲义

下册

东南大学高等数学教研室

2020 年 2 月

实验一　　空间曲线与曲面的绘制

本实验的目的是利用数学软件Mathematica绘制三维图形来观察空间曲线和空间曲面图形的特点。

1、空间曲线的绘制

绘制空间曲线时一般使用曲线的参数方程，利用命令“**ParametricPlot3D**”。如画出参数方程 $\begin{cases} x = x(t) \\ y = y(t), & t_1 \le t \le t_2 \\ z = z(t) \end{cases}$ 所确定的空间曲线的命令格式为:

ParametricPlot3D[{x[t],y[t],z[t]},{t,tmin,tmax},选项]

例1　画出旋转抛物面 $z = x^2 + y^2$ 与上半球面 $z = 1 + \sqrt{1 - x^2 - y^2}$ 交线的图形。

解：它们的交线为平面 $z = 1$ 上的圆 $x^2 + y^2 = 1$，化为参数方程为 $\begin{cases} x = \cos t \\ y = \sin t, & t \in [0, 2\pi], \\ z = 1 \end{cases}$

下面的 mathematica 命令就是作出它们的交线并把它存在变量 p 中：

```
In[1]:= p = ParametricPlot3D[{Cos[t], Sin[t], 1}, {t, 0, 2*Pi}]
```

运行即得曲线如图 1-1 所示。

在这里说明一点，要作空间曲线的图形，必须先求出该曲线的参数方程。如果曲线为一般式 $\begin{cases} F(x,y,z) = 0 \\ G(x,y,z) = 0 \end{cases}$，其在 xOy 面上的投影柱面的准线方程为 $H(x,y) = 0$，可先将 $H(x,y) = 0$ 化为参数方程 $\begin{cases} x = x(t) \\ y = y(t) \end{cases}$，再代入 $G(x,y,z) = 0$ 或 $F(x,y,z) = 0$ 解出 $z = z(t)$ 即可。



图 1-1



图 1-2

2、 空间曲面的绘制

作一般式方程 $z = f(x, y)$ 所确定的曲面图形的 Mathematica 命令为：

```
Plot3D[f[x,y],{x,xmin,xmax},{y,ymin,ymax},选项]
```

作参数方程 $\begin{cases} x = x(u,v) \\ y = y(u,v), u \in [u_{\min}, u_{\max}], v \in [v_{\min}, v_{\max}] \\ z = z(u,v) \end{cases}$ 所确定的曲面图形的 Mathematica 命令为：

```
ParametricPlot3D[{x[u,v],y[u,v],z[u,v]},{u,umin,umax},
{v,vmin,vmax},选项]
```

绘制三维图形（空间曲线、空间曲面）常用的选项见简介。

例2　作出上半球面 $z = 1 + \sqrt{1 - x^2 - y^2}$ 的图形。

解：首先我们选取绘图区间 $\{-1 < x < 1, -1 < y < 1\}$ 作图，输入下面语句：

```
In[2]:= Plot3D[1 + Sqrt[1 - x^2 - y^2], {x, -1, 1}, {y, -1, 1}]
```

运行后得到了该曲面的图形（图 1-2）。

我们可以定义一个分区域函数：

$$f(x, y) = \begin{cases} 1 + \sqrt{1 - x^2 - y^2}, & x^2 + y^2 \le 1 \\ 1, & x^2 + y^2 > 1 \end{cases},$$

来作出该函数的图形。键入命令：

```
In[3]:= f[x_, y_] := If[x^2 + y^2 ≤ 1, 1 + Sqrt[1 - x^2 - y^2], 1]
        Plot3D[f[x, y], {x, -1, 1}, {y, -1, 1}]
```

运行后得图 1-3，可以看到该图形比上半球面多了一部分曲面的图形（即 $z = 1$ 平面上的部分）。

对球面，我们也可以采用参数方程来绘图。选取参数的范围使得区域内的每一点都有定义。对于题目中的球面有参数方程 $\begin{cases} x = \cos u \sin v \\ y = \sin u \sin v, \ u \in [0, 2\pi], v \in [0, \frac{\pi}{2}] \\ z = 1 + \cos v \end{cases}$，我们输入命令：

```
In[5]:= ParametricPlot3D[{Cos[u] * Sin[v], Sin[u] * Sin[v],
    1 + Cos[v]}, {u, 0, 2 * Pi}, {v, 0, Pi/2}, PlotPoints → 30]
```

运行后输出图形，如图 1-4.



图 1-3



图 1-4

我们还可以改变参数的范围画出上半球面的$\frac{3}{4}$部分（如图 1-5）:

```
In[6]:= ParametricPlot3D[{Cos[u] * Sin[v], Sin[u] * Sin[v],
    1 + Cos[v]}, {u, 0, 3 * Pi / 2}, {v, 0, Pi / 2},
   PlotPoints → 30]
```



图 1-5

3、 空间图形的叠加

与平面图形类似，空间的立体图形同样可用“Show”命令，把不同的图形（曲线或曲面）叠加并在一个坐标系中显示出来。

例 3 画出由旋转抛物面 $z = x^2 + y^2$ 与上半球面 $z = 1 + \sqrt{1 - x^2 - y^2}$ 相交所围成的立体几何图形。

解：这是一个组合图形。一般地，直接画出两者的图形再组合在一起。但是，这里所要的图形仅仅是两个曲面图形的一部分，因此需要有选择地画出两曲面的相应部分再组合。由于它们的交线为$\begin{cases} x^2+y^2=1 \\ z=1 \end{cases}$，故相应的曲面部分的参数方程为：

$$\begin{cases} x=r\cos t \\ y=r\sin t,\ t\in[0,2\pi], r\in[0,1] \\ z=r^2 \end{cases} \text{与} \begin{cases} x=\cos u\sin v \\ y=\sin u\sin v,\ u\in[0,2\pi], v\in[0,\dfrac{\pi}{2}] \\ z=1+\cos v \end{cases}\text{。}$$

输入以下 Mathematica 语句：

```
In[7]:= t1 = ParametricPlot3D[{r*Cos[t], r*Sin[t], r^2},
            {t, 0, 2*Pi}, {r, 0, 1}, PlotPoints → 30];
        t2 = ParametricPlot3D[
            {Cos[r]*Sin[t], Sin[r]*Sin[t], 1+Cos[t]},
            {r, 0, 2*Pi}, {t, 0, Pi/2}, PlotPoints → 30];
        tu = Show[t1, t2, PlotRange → All]
```

运行后即得旋转抛物面、上半球面及叠加曲面的图形（图 1-6）。



图 1-6

4、 常见的二次曲面的参数方程

椭球面 $\begin{cases} x=a\sin u\sin v, \\ y=b\sin u\cos v, \quad u\in[0,\pi], v\in[0,2\pi] \\ z=c\cos u, \end{cases}$

椭圆抛物面（部分）$\begin{cases} x = au\sin v, \\ y = bu\cos v, \quad u \in [0,r], v \in [0,2\pi] \\ z = u^2, \end{cases}$

双曲抛物面（部分）$\begin{cases} x = u, \\ y = v, \quad u \in [-r,r], v \in [-r,r] \\ z = \dfrac{u^2}{a} - \dfrac{v^2}{b}, \end{cases}$

单叶双曲面（部分）$\begin{cases} x = a\sec u\sin v, \\ y = b\sec u\cos v, \quad u \in [-\dfrac{\pi}{4}, \dfrac{\pi}{4}], v \in [0,2\pi] \\ z = c\tan u, \end{cases}$

双叶双曲面（部分）$\begin{cases} x = a\sqrt{u^2-1}\sin v, \\ y = b\sqrt{u^2-1}\cos v, \quad |u| \in [1,5], v \in [0,2\pi] \\ z = cu, \end{cases}$

圆锥面（部分）$\begin{cases} x = u\cos v, \\ y = u\sin v, \quad u \in [-a,a], v \in [0,2\pi] \\ z = u, \end{cases}$

上面各式或区间中的字母均表示正数，在作图的过程中大家可以自行选择。下面用动画来演示产生旋转曲面的过程。

例3　用动画演示由曲线 $y = \sin z, z \in [0,\pi]$ 绕 z 轴旋转产生旋转曲面的过程。

解：该曲线绕绕 z 轴旋转产生的曲面方程为 $x^2 + y^2 = \sin^2 z$，其参数方程为 $\begin{cases} x = \sin z\cos u \\ y = \sin z\sin u, \\ z = z \end{cases}$ $z \in [0,\pi], u \in [0,2\pi]$，输入以下命令，就可得到连续变化的 20 幅图形：

```
In[10]:= m = 20;
For[i = 1, i ≤ m, i++,
  a = ParametricPlot3D[{Sin[z] * Cos[u], Sin[z] * Sin[u], z},
    {z, 0, Pi}, {u, 0, 2 Pi * i / m}, AspectRatio → 1,
    AxesLabel → {"X", "Y", "Z"}, PlotPoints → 30];
  Print[a]];
```

运行后得到 20 幅曲面的图形，图 1-7 中列举了其中的六幅，大家可以改变循环语句中的参数，画出更多的图形，从而观察到旋转曲面产生的过程。



图 1-7

实验习题一

1、 作出各种标准二次曲面的图形。

2、 利用参数方程作图，作出由下列曲面所围成的立体：

（1） $z=\sqrt{1-x^2-y^2}$、$x^2+y^2=x$ 及 xOy 面

（2） $z=xy$、$x+y-1=0$ 及 $z=0$

3、观察二次曲面族 $z=x^2+y^2+kxy$ 的图形。特别注意确定 k 的这样一些值，当 k 经过这些值时，曲面从一种类型变成了另一种类型。

实验二 最小二乘法

在科学研究和实际工作中，常常会遇到这样的问题：给定两个变量 x, y 的 m 组实验数据 $(x_1, y_1), (x_2, y_2), \cdots, (x_m, y_m)$，如何从中找出这两个变量间的函数关系的近似解析表达式（也称为经验公式），使得能对 x 与 y 之间的除了实验数据外的对应情况作出某种判断。

这样的问题一般可以分为两类：一类是对要对 x 与 y 之间所存在的对应规律一无所知，这时要从实验数据中找出切合实际的近似解析表达式是相当困难的，俗称这类问题为黑箱问题；另一类是依据对问题所作的分析，通过数学建模或者通过整理归纳实验数据，能够判定出 x 与 y 之间满足或大体上满足某种类型的函数关系式 $y = f(x, a)$，其中 $a = (a_1, a_2, \cdots, a_n)$ 是 n 个待定的参数，这些参数的值可以通过 m 组实验数据来确定（一般要求 $m > n$），这类问题称为灰箱问题。解决灰箱问题的原则通常是使拟合函数在 x_i 处的值与实验数值的偏差平方和最小，即 $\sum_{i=1}^{n}[f(x_i, a) - y_i]^2$ 取得最小值。这种在方差意义下对实验数据实现最佳拟合的方法称为“最小二乘法”，$a_1, a_2, \cdots, a_n$ 称为最小二乘解，$y = f(x, a)$ 称为拟合函数。下面对于“最小二乘法”通过两个例子来说明。

一、线性拟合问题

例 1．在某化工厂生产过程中，为研究温度 x（单位：摄氏度）对收率（产量）y (%)的影响，可测得一组数据如下表所示，试根据这些数据建立以 x 与 y 之间的拟合函数。

温度 x	100	110	120	130	140	150	160	170	180	190
收率 y(%)	45	51	54	61	66	70	74	78	85	89

解：（1）确定函数的类型

为此，应对给定的数据以 x 为横坐标，y 为纵坐标作出散点图。输入如下语句：

```
In[1]:= x = Table[100 + 10 * i, {i, 0, 9}];
        y = {45, 51, 54, 61, 66, 70, 74, 78, 85, 89};
        xy = Table[{x[[i]], y[[i]]}, {i, 1, 10}]
        ListPlot[xy, PlotStyle → PointSize[0.015]]
```

运行后可得到数据表（图 2-1）和图 2-2。

从图 2-1 种可以看出这些点近似地落在一条直线周围，可以认为在 y 和 x 之间存在着线

性关系，之所以不完全落在直线上，是因为观察数据本身存在的误差。下面我们就用最小二乘法求出与这些数据点最接近的直线方程。

```
Out[3]= {{100, 45}, {110, 51},
         {120, 54}, {130, 61},
         {140, 66}, {150, 70},
         {160, 74}, {170, 78},
         {180, 85}, {190, 89}}
```

图 2-1



图 2-2

（2）利用 Mathmatica 求最小二乘解

设直线方程式为 $y = ax + b$，其中 a, b 是待定系数，于是，可以把拟合函数在 x_i 处的值与观察值的偏差平方和表示为关于待定系数 a, b 的二元函数：

$$Q(a,b)=\sum_{i=1}^{n}[(ax_i+b)-y_i]^2\text{，}$$

按照二元函数求极值的理论，其最小值应满足方程组：

$$\begin{cases}\dfrac{\partial Q}{\partial a}=\sum\limits_{i=1}^{n}2[(ax_i+b)-y_i]x_i=0\\ \dfrac{\partial Q}{\partial b}=\sum\limits_{i=1}^{n}2[(ax_i+b)-y_i](-1)=0\end{cases}\text{，即}\begin{cases}\sum\limits_{i=1}^{n}[(ax_i+b)-y_i]x_i=0\\ \sum\limits_{i=1}^{n}[(ax_i+b)-y_i]=0\end{cases}\text{。}$$

于是，输入以下 Mathematica 语句并求解待定系数 a, b：

```
In[5]:= q[a_, b_] := Sum[(a x[[i]] + b - y[[i]])^2, {i, 1, 10}]
        NSolve[{D[q[a, b], a] == 0, D[q[a, b], b] == 0}, {a, b}]

Out[6]= {{a → 0.48303, b → -2.73939}}
```

从运行结果可出看出 $a = 0.48303,\ b = -2.73979$，这样，我们得到方程 $y = 0.48303x - 2.73979$，这就是前面所说的经验公式（注：如果在 In[5]中将命令 NSolve 改为 Solve 命令，则得到精确解 $a=\dfrac{797}{1650},\ b=-\dfrac{452}{165}$）。

在此例题中，观察的数据点大致呈一直线状，如果在实际问题中数据点落在某一条曲线周围，则要根据曲线的形状来确定拟合函数的类型。比如考虑用 x 的 n 次多项式 $P_n(x)=a_0+a_1x+a_2x^2+\cdots+a_nx^n$ 来拟合，这成为多项式拟合。此时仍可用最小二乘法，

考虑 $n+1$ 元函数 $Q(a_0,a_1,a_2,\cdots,a_n)=\sum_{i=1}^{n}[P_n(x_i)-y_i]^2$ 取最小的必要条件，令此函数对各个参数的偏导等于 0，解一个 $n+1$ 元的方程组便可求得这些参数的最小二乘解。

二、可化为线性拟合的曲线拟合问题

在许多场合下，拟合函数不具有线性形式，但是由实际经验或相关的学科理论，能够提供拟合函数的可取类型，而且可以通过适当的变量代换将拟合函数线性化，同样可以建立经验公式。

（1）模型 $y=ae^{bx}$ 可以用变量替换 $Y=\ln y, X=x$ 将函数化为线性函数：$Y=\ln a+bX$ 。

（2）模型 $y=a+b\ln x$ 可以用变量替换 $Y=y, X=\ln x$ 将函数化为线性函数。

（3）模型 $\varphi(y)=a+bx$ 可以用变量替换 $z=\varphi(y)$ 将其化为 z 和 x 之间的线性拟合。

例 2．研究黏虫的生长过程，可测的一组数据下表所示。

温度 t	11.8	14.7	15.4	16.5	17.1	18.3	19.8	20.3
历期 N	30.4	15	13.8	12.7	10.7	7.5	6.8	5.7

其中历期 N 是指卵块孵化成幼虫的天数。昆虫学家认为在 N 与 t 之间有关系式：$N=\dfrac{k}{t-c}$，其中 k,c 为常数，试求最小二乘解。

解：作变换 $y=\dfrac{1}{N},x=t,a=\dfrac{1}{k},b=-\dfrac{c}{k}$，由此把 N,t 的关系式化为了关于 x,y 的线性关系式：$y=\dfrac{t-c}{k}=\dfrac{1}{k}x--\dfrac{c}{k}=ax+b$ 。与例 1 相同，输入以下语句并求解参数 a,b:

```
In[9]:= t = {11.8, 14.7, 15.4, 16.5, 17.1, 18.1, 19.8, 20.3};
        w = {30.4, 15.0, 13.8, 12.7, 10.7, 7.5, 6.8, 5.7};
        y = 1 / w;
        x = t;
        xy = Table[{x[[i]], y[[i]]}, {i, 1, 8}];
        q[a_, b_] := Sum[(a x[[i]] + b - y[[i]])^2, {i, 1, 8}]
        Solve[{D[q[a, b], a] == 0, D[q[a, b], b] == 0}, {a, b}]

Out[15]= {{a → 0.016481, b → -0.175433}}
```

从运行后的结果看出 $a=0.016481,\ b=-0.175433$ 。最后为了求出参数 k,c 以及经验公式，还需要回代变量，为此输入命令并运行：

```
In[16]:= A = {a, b} /. %;
         k = 1 / A[[1, 1]]
         c = -k * A[[1, 2]]

Out[17]= 60.6758

Out[18]= 10.6445
```

从运行结果得到参数 k, c 的值分别为 $k = 60.6758,\ c = 10.6445$，由此得到拟合曲线方程为：

$$N = \frac{60.6758}{t - 10.6445}。$$

为了比较得到的拟合函数和已知的数据点，我们再在同一坐标下绘出数据点的散点图及拟合函数的图形，输入语句为：

```
In[19]:= data = Table[{t[[i]], w[[i]]}, {i, 1, 8}];
         t1 = ListPlot[data, PlotStyle → PointSize[0.015]];
         f[x_] := k / (x - c);
         t2 = Plot[f[x], {x, 10.7, 25}];
         Show[t1, t2, PlotRange → {0, 100}]
```

运行结果如图 2-3 所示。



图 2-3

实验习题二

1．为测定刀具的磨损速度，每隔一小时测量一次刀具的厚度，由此得到以下数据：

时间 t	0	1	2	3	4	5	6	7
厚度 y	27.0	26.8	26.5	26.3	26.1	25.7	25.3	24.8

试根据这组数据建立 y 与 t 之间的拟合函数。

2．一种合金在某种添加剂的不同浓度下进行实验，得到如下数据：

浓度 x	10.0	15.0	20.0	25.0	30.0
抗压强度 y	27.0	26.8	26.5	26.3	26.1

已知函数 y 与 x 的关系适合模型： $y = a + bx + cx^2$，试用最小二乘法确定系数 a,b,c，并求出拟合曲线。

3．在研究化学反应速度时，得到下列数据：

x_i	3	6	9	12	15	18	21	24
y_i	57.6	41.9	31.0	22.7	16.6	12.2	8.9	6.5

其中 x_i 表示实验中作记录的时间， y_i 表示在相应时刻反应混合物中物质的量，试根据这些数据建立经验公式。

实验三　重积分的数值计算

我们知道，重积分是定积分的推广，它可以通过表示成累次积分来计算，这说明重积分的数值计算方法可以基于定积分的数值计算方法来设计。本实验的目的是通过几个具体例子来介绍使用 Mathematica 计算重积分。在实验中，借助于 Mathematica 对图形进行分析，来确定累次积分的积分限。

1、 重积分的计算方法

例1　计算下列二重积分：

（1）$\iint\limits_D e^{-x^2}dxdy$，其中 D 是矩形区域 $|x|\le 1, |y|\le 1$。

（2）$\iint\limits_D \sin x^2 dxdy$，其中 D 是由直线 $y=x^2$ 和 $y=x+6$ 围成的区域。

解：（1）$\iint\limits_D e^{-x^2}dxdy=\int_{-1}^{1}dy\int_{-1}^{1}e^{-x^2}dx$，故由下面命令：

```
In[1]:= NIntegrate[Exp[-x^2], {y, -1, 1}, {x, -1, 1}]
```

或者由基本输入模板输入命令：

$$\text{In[2]:= } \mathbf{N}\left[\int_{-1}^{1}\int_{-1}^{1}\mathbf{Exp}[-x\,^\wedge 2]\, dx\, dy\right]$$

运行后均可得到积分值：2.9873。

（2）易知 $\iint\limits_D \sin x^2 dxdy=\int_{-2}^{3}dx\int_{x^2}^{x+6}\sin x^2 dy$，故键入如下连个命令：

```
In[3]:= NIntegrate[Sin[x^2], {x, -2, 3}, {y, x^2, x + 6}]
```

$$\mathbf{N}\left[\int_{-2}^{3}\int_{x^2}^{x+6}\mathbf{Sin}[x\,^\wedge 2]\, dy\, dx\right]$$

均可得到所求积分值：6.99628。

这里要注意的是函数“`NIntegrate`”中的积分次序是按照书写的先后次序，由基本输入模板得到的命令中，dx 和 dy 的顺序也不能调换，它是一个先 y 后 x 的积分。

2、 空间图形分析与投影区域的确定

在三重积分中，积分区域是立体图形，我们可以利用软件 Mathematica 画出积分区域图形，来帮助确定累次积分的上下限。

绘制曲面 $\begin{cases} x = x(u,v) \\ y = y(u,v),\ u \in [u_{\min}, u_{\max}], v \in [v_{\min}, v_{\max}] \\ z = z(u,v) \end{cases}$ 在 yOz 坐标面上的投影时，将 x 设为常数 a，命令如下：

```
ParametricPlot3D[{a, y(u,v), z(u,v)},
                 {u, umin, umax},{v, vmin, vmax},选项]
```

同理可得绘制曲面在 xOy、xOz 坐标面上的投影的命令。

例2 求三重积分 $\iiint_{\Omega} xy^2z^2dV$，其中 Ω 是由曲面 $z = x^2 + y^2$、$x = y^2$、$x = 1$ 与 $z = 0$ 所围成的立体区域。

解：首先作出立体区域 Ω 的图形。先分别作出各曲面的图形，再在同一坐标系下显示它们的图形，输入命令为：

```
In[5]:= s1 = ParametricPlot3D[{u, v, u^2 + v^2}, {u, -1, 1},
          {v, -1, 1}, PlotPoints → 50, AxesLabel → {"X", "Y", "Z"}];
        s2 = ParametricPlot3D[{u^2, u, v}, {u, -1, 1}, {v, 0, 2},
          AxesLabel → {"X", "Y", "Z"}];
        s3 = ParametricPlot3D[{1, u, v}, {u, -1, 1}, {v, 0, 2},
          AxesLabel → {"X", "Y", "Z"}];
        s4 = ParametricPlot3D[{u, v, 0}, {u, -1, 1}, {v, -1, 1},
          AxesLabel → {"X", "Y", "Z"}];
        Show[s1, s2, s3, s4]
```

运行后得到立体区域 Ω 如图 3-1 所示。从图中可以看出，该区域是以 $z = x^2 + y^2$ 为顶、以 $z = 0$ 为底，以 $x = y^2$、$x = 1$ 为侧面的一个立体。下面要求立体在 xOy 面的投影区域。



图 3-1



图 3-2

为此，我们将曲面$z=x^2+y^2$、$x=y^2$、$x=1$在xOy面的投影在同一坐标下显示，输入命令如下：

```
In[9]:= t1 = ParametricPlot3D[{u, v, 0}, {u, -1, 1}, {v, -1, 2},
          AxesLabel → {"X", "Y", "Z"}];
        t2 = ParametricPlot3D[{u^2, u, 0}, {u, -1, 1},
          AxesLabel → {"X", "Y", "Z"}];
        t3 = ParametricPlot3D[{1, u, 0}, {u, -1, 1},
          AxesLabel → {"X", "Y", "Z"}];
        Show[t1, t2, t3, PlotRange → {0, 1}]
```

运行后得到图 3-2。从图中可以看出，Ω在xOy面的投影域为$x=y^2$、$x=1$所构成的平面区域。故$\iiint\limits_{\Omega} xy^2z^2dxdydz=\int_{-1}^{1}dy\int_{y^2}^{1}dx\int_{0}^{x^2+y^2}xy^2z^2dz$。于是输入以下两条命令并运行：

```
In[13]:= ∫_{-1}^{1} ∫_{y^2}^{1} ∫_{0}^{x^2+y^2} x y^2 z^2 dz dx dy
         Integrate[x y^2 z^2, {y, -1, 1}, {x, y^2, 1}, {z, 0, x^2 + y^2}]
Out[13]= 475936/3968055
Out[14]= 475936/3968055
```

均可得到所求三重积分的值：$\dfrac{475936}{3968055}$。

例 3 求三重积分$\iiint\limits_{\Omega} zdV$，其中$\Omega$是由曲面$x^2+y^2+z^2=4\ (z\geq 0)$与$z=\dfrac{1}{3}(x^2+y^2)$所围成的立体区域。

解：首先作出区域Ω的图形，输入命令如下：

```
In[15]:= g1 = ParametricPlot3D[{2 Sin[u] Cos[v], 2 Sin[u] Sin[v], 2 Cos[u]},
           {u, 0, Pi / 2}, {v, 0, 2 Pi}, PlotRange → {-2, 2.5}];
         g2 = ParametricPlot3D[{1/√3 u Cos[v], 1/√3 u Sin[v], 1/3 u^2},
           {u, -3, 3}, {v, 0, 2 Pi}, PlotRange → {-2, 2.5}];
         Show[g1, g2]
```

运行后即得图 3-3。

从图中可以看出这个区域以$z=\sqrt{4-x^2-y^2}$为顶、以$z=\dfrac{1}{3}(x^2+y^2)$为底的一个立

体，它在 xOy 面上的投影即为这两个曲面的交线在 xOy 面上的投影曲线所围成的区域。我们可以将两曲面在 xOy 面的投影显示在同一坐标系中，由此看出投影区域为一个圆域。下面来求投影曲线，为此利用命令“Eliminate”消去两曲面中的变量 z，即键入命令并运行如下：



图 3-3

```
In[18]:= Eliminate[{z == Sqrt[4 - x^2 - y^2], z^2 == 1/3 (x^2 + y^2)}, z]
Out[18]= 3 - y^2 == x^2
```

由此看出投影曲线方程为：$x^2+y^2=3$，故 $\iiint\limits_{\Omega} zdxdydz=\int_{-\sqrt{3}}^{\sqrt{3}}dy\int_{-\sqrt{3-y^2}}^{\sqrt{3-y^2}}dx\int_{\sqrt{4-x^2-y^2}}^{\frac{1}{3}(x^2+y^2)}zdz$，

于是运行以下两条命令：

```
In[19]:= ∫_{-√3}^{√3} ∫_{-√(3-y^2)}^{√(3-y^2)} ∫_{√(4-x^2-y^2)}^{1/3 (x^2+y^2)} z dz dx dy
Integrate[z, {y, -Sqrt[3], Sqrt[3]},
 {x, -Sqrt[3 - y^2], Sqrt[3 - y^2]},
 {z, Sqrt[4 - x^2 - y^2], 1 / 3 (x^2 + y^2)}]
```

均可得到所求积分值：$-\dfrac{13}{4}\pi$。

实验习题三

作出下列积分区域的图形，并计算积分：

1、$\iint\limits_{D} xydxdy$，其中 D 是由 $y^2=x$ 与 $y=x-2$ 围成的区域。

2、$\iiint\limits_{\Omega}(x^2+y^2)dV$，其中 Ω 是由曲面 $z=e^{-x^2-y^2}$、$z=0$、$x^2+y^2=4$ 所围成的立体区域。

3、$\iiint\limits_{\Omega} x^2y^2zdV$，其中 Ω 是由曲面 $z=xy$、$x+y-1=0$ 及 $z=0$ 所围成的立体区域。

实验四　无穷级数与函数逼近

本实验的目的是用 Mathematica 显示级数部分和的变化趋势；学会如何利用幂级数的部分和对函数的逼近以及进行函数值的近似计算；展示傅里叶级数对周期函数的逼近情况。

1、 级数部分和的变化趋势

在实验一中，我们通过图形能够清楚地显示极限的变化趋势，而级数的和就是部分和序列的极限，下面我们采用散点图来观察级数部分和序列的变化趋势。

例 1 观察级数 $\sum_{n=1}^{\infty}\frac{(-1)^{n-1}}{2n-1}$ 的部分和序列的变化趋势，并求和。

解：（1）可以用以下两种方法，从图形来观察级数的敛散性。

（a）利用“Table”命令生成部分和数列的数据点集后作点图，输入语句如下：

```
In[1]:= s[n_] := Sum[(-1)^(k-1)/(2k-1), {k, 1, n}];
        data = Table[s[n], {n, 1, 400}];
        ListPlot[data]
```





图 4-1　　　　　　　　　　　　图 4-2

运行后见图 8-1。从图中可以看到级数收敛，级数和大约为 0.786。

（b）将级数的所有部分和用竖直线段画出，得到类似条形码的图形，通过这种图形来看出级数的收敛性。输入如下命令：

```
In[3]:= sn = 0; n = 1; h = {}; m = 3;
        While[1/n > 10^(-m), sn = sn + (-1)^(n-1)/(2n-1);
          h =
           Append[h,
            Graphics[{RGBColor[Abs[Sin[n]], 0, 1/n],
              Line[{{sn, 0}, {sn, 1}}]}]]; n++];
        Show[h, PlotRange → {0.1, 0.103}, Axes → True]
```

运行后见图 4-2。从图中可以看出，级数的和在 0.785 与 0.786 之间，并且可以通过改变 m 的值来提高观察到的和的精度。

（2）求和。

对于这个级数，可以通过基本输入模板利用求和符号来直接求和，只要输入如下命令并运行：

```
In[5]:= Sum[(-1)^(n-1)/(2n-1), {n, 1, Infinity}]
```

$$\sum_{n=1}^{\infty}\frac{(-1)\,\hat{}\,(n-1)}{2n-1}$$

Out[5]= $\frac{\pi}{4}$

得到级数和的精确值 $\frac{\pi}{4}$。但并不是所有的级数都能如此求得和，我们可以利用下面的两个命令来求和的近似值：

In[6]:= $\sum_{n=1}^{\infty}\frac{(-1)\,\hat{}\,(n-1)}{2n-1}$ // N

NSum$\left[\frac{(-1)\,\hat{}\,(n-1)}{2n-1}\text{, \{n, Infinity\}}\right]$

运行后得结果均为：0.785398。为了得到更高的精度，还可以用“N”命令来求函数“Sum”所得的值。大家可以比较一下下面两条命令输出结果的差异：

In[8]:= N$\left[\text{Sum}\left[\frac{(-1)\,\hat{}\,(n-1)}{2n-1}\text{, \{n, Infinity\}}\right]\text{, 30}\right]$

N$\left[\text{NSum}\left[\frac{(-1)\,\hat{}\,(n-1)}{2n-1}\text{, \{n, Infinity\}}\right]\text{, 30}\right]$

2、 函数的幂级数展开

如果函数 $f(x)$ 在 x_0 邻域内具有任意阶导数，则函数可以展开为 x_0 处的幂级数

$$f(x)=\sum_{n=1}^{\infty}a_n(x-x_0)^n=\sum_{n=1}^{\infty}\frac{f^{(n)}(x_0)}{n!}(x-x_0)^n,$$

称之为泰勒级数。特别地，当 $x_0=0$ 时，称为麦克劳林级数。

例2　将函数 $f(x)=(1+x)^m$ 展开为 x 的幂级数，并利用图形考察幂级数的部分和逼近函数的情况。

解：根据幂级数的展开公式，若 $f(x)$ 能展开成 x 的幂级数，其展示为 $f(x)=\sum_{n=1}^{\infty}\frac{f^{(n)}(0)}{n!}x^n$，

因此首先定义函数，再计算 $x=0$ 点的 n 阶导数，最后构成和式。不妨设 $m=-1$，输入命令如下：

```
In[10]:= m = -1; f[x_] := (1 + x)^m; x0 = 0;
         g[n_, x0_] := D[f[x], {x, n}] /. x → x0;
         s[n_, x_] := Sum[g[k, x0]/k! * (x - x0)^k, {k, 0, n}];
```

命令中的函数 **s[n_,x_]**表示的是函数 $f(x)$ 在 x_0 处的 n 阶泰勒级数。下面我们通过以下命令观察幂级数的部分和逼近函数的情况：

```
In[13]:= t = Table[s[n, x], {n, 20}];
         p1 = Plot[Evaluate[t], {x, -1/2, 1/2}];
         p2 = Plot[(1 + x)^m, {x, -1/2, 1/2},
            PlotStyle → RGBColor[1, 0, 0]];
         Print[p1];
         Print[p2];
         Show[p1, p2]
```

运行结果如图 4-3。从图中可以看到，当 n 越大时，幂级数越逼近函数。



图 4-3

3、傅里叶级数

设 $f(x)$ 是以 2T 为周期的周期函数，在任一周期内，$f(x)$ 除在有限个第一类间断点外都连续，并且只有有限个极值点，则 $f(x)$ 可以展开为傅里叶级数：

$$\frac{a_0}{2}+\sum_{n=1}^{\infty}(a_n\cos\frac{n\pi x}{T}+b_n\sin\frac{n\pi x}{T}),\text{ 其中}\begin{cases}a_n=\dfrac{1}{T}\displaystyle\int_{-T}^{T}f(x)\cos\frac{n\pi x}{T}dx, & n=0,1,2,\cdots\\ b_n=\dfrac{1}{T}\displaystyle\int_{-T}^{T}f(x)\sin\frac{n\pi x}{T}dx, & n=1,2,3,\cdots\end{cases},$$

且傅里叶级数在任一点 x_0 处收敛于 $\dfrac{f(x_0-0)+f(x_0+0)}{2}$。

下面通过例子来研究傅里叶级数的生成以及级数部分和逼近函数的情况。

例3　画图观察周期为2π、振幅为 1 的方波函数 $f(x)=\begin{cases}-1, & -\pi\le x<0\\ 1, & 0\le x<\pi\end{cases}$ 展成的傅里叶级数的部分和逼近 $f(x)$ 的情况。

解：根据傅里叶系数公式可得：$a_0=\frac{1}{\pi}\int_{-\pi}^{\pi}f(x)dx=0$，

$$a_n=\frac{1}{\pi}[\int_{-\pi}^{0}(-\cos nx)dx+\int_0^{\pi}\cos nxdx]，\quad b_n=\frac{1}{\pi}[\int_{-\pi}^{0}(-\sin nx)dx+\int_0^{\pi}\sin nxdx],$$

故输入以下命令，从输出的图形（图 4-4）观察傅里叶级数的部分和逼近 $f(x)$ 的情况：

```
In[19]:= f[x_] := Which[-2 Pi ≤ x < -Pi, 1, -Pi ≤ x < 0, -1, 0 ≤ x < Pi, 1,
            Pi ≤ x < 2 Pi, -1];
         a[n_] :=
           {Integrate[-Cos[n x], {x, -Pi, 0}] + Integrate[Cos[n x], {x, 0, Pi}]} / Pi;
         b[n_] :=
           {Integrate[-Sin[n x], {x, -Pi, 0}] + Integrate[Sin[n x], {x, 0, Pi}]} / Pi;
         s[x_, n_] := a[0]/2 + Sum[a[k] * Cos[k x] + b[k] * Sin[k x], {k, 1, n}];
         g1 = Plot[f[x], {x, -2 Pi, 2 Pi}, PlotStyle → RGBColor[0, 0, 1]];
         m = 18;
         For[i = 1, i ≤ m, i += 2, t = Evaluate[s[x, i]];
          g2 = Plot[{f[x], t}, {x, -2 Pi, 2 Pi},
            PlotStyle → {RGBColor[0, 0, 1], RGBColor[1, 0, 0]}];
          Print[g2]]
```



图 4-4

从图表可以看出，n 越大逼近函数的效果越好，还可以注意到傅里叶级数的逼近是整体性的。

例4　将函数 $f(x)=x^4+1$ 展开成周期为 3 的傅里叶级数。

解：输入命令如下：

```
In[26]:= fourier[f_, T_, k_] := Module[{a, b, i, t, s, g1, g2},
    a[0] = Integrate[f, {x, -T, T}] / T; s = a[0] / 2;
    For[i = 1, i ≤ k, i++, t = i * Pi / T;
     a[i_] := Integrate[f * Cos[t * x], {x, -T, T}] / T;
     b[i_] := Integrate[f * Sin[t * x], {x, -T, T}] / T;
     s = s + a[i] Cos[t * x] + b[i] Sin[t * x]]; Print[s];
    Plot[Evaluate[s], {x, -T, T}]];
  f = x^4 + 1; T = 3; n = 8;
  g = Plot[f, {x, -T, T}, PlotStyle → RGBColor[1, 0, 0]];
  Print[g];
  For[j = 1, j ≤ n, j += 2, p = fourier[f, T, j]];
  Show[g, p]
```

在上面的程序中，先定义了一个产生傅里叶级数的前 k 项部分和的函数，在后面的“For”循环中连续调用该函数，输出了 $f(x)$ 的傅里叶级数的前 $n(n=1,3,5,7,)$ 项部分和函数，并画出了 $f(x)$ 图形及同一坐标下 $f(x)$ 与其傅里叶级数前 7 项和函数的图形（如图 4-5）。



图4-5

实验习题四

1、 观察级数 $\sum_{n=1}^{\infty}\frac{n!}{n^n}$ 的部分和序列的变化趋势，并求和

2、 改变例2中m及 x_0 的数值来求函数的幂级数及观察其幂级数的逼近函数的情况。

3、 观察函数 $f(x)=\begin{cases}-x, & -\pi\le x<0\\ 1, & 0\le x<\pi\end{cases}$ 展成的傅里叶级数的部分和逼近 $f(x)$ 的情况。

实验五　特殊图形

在前面的实验中，我们介绍了 Mathematica 中用于绘制二维和三维图形的绘图函数“Plot”、“Plot3D”、“ParametricPlot”、“ParametricPlot3D”，除了这几个函数外，Mathematica 还提供了绘制一些特殊图形的函数。在本实验中，将给大家介绍二元函数的等高线和梯度线的绘制，以及一元隐函数图形的画法。

1、 等高线

二元函数可视化有两种方式：

（1）在 R^3 中画出它的图形，即画出点集 $\{(x,y,z)\,|\,z=f(x,y)\}$ 的图形；

（2）在 R^2 中画出它的等高线，即曲面 $z=f(x,y)$ 与平面 $z=C$ 相交的曲线。

等高线也称为等值线。等高线具有以下特点：高度相同的点在同一等高线上；等高线图上亮度越高的地方，所对应曲面在相应之处的值也就越高。在 Mathematica 中，绘制等高线命令为：“**ContourPlot**”，它调用的格式为：

ContourPlot[f[x,y],{x，xmin,xmax},{y,ymin,ymax},选项]

例 1 绘制函数 $f(x,y)=\dfrac{2x^3}{x^2+y^2}$ 的等高线图。

解：为了比较该函数所表示的曲面和等高线之间的关系，运行下面的语句后，输出了函数 $z=f(x,y)$ 的图形（图 5-1）及其等高线的图形（图 5-2）。

```
In[1]:= f[x_, y_] := 3 x^3 / (x^2 + y^2);
        Plot3D[f[x, y], {x, -1, 1}, {y, -1, 1}]
        ContourPlot[f[x, y], {x, -1, 1}, {y, -1, 1}]
```



图 5-1



图 5-2

这里所得的等高线图是以最基本的等高线作图命令作出的，选项均采用了默认状态，此时的等高线数为 4 条，这可能使某些细节没有表现出来。我们可以在命令中添加选项来对图形作适当的调整。例如要规定等高线的条数、提高采样点数、规定作图范围可键入命令：

```
ContourPlot[f[x,y],{x,-1,1},{y,-1,1},Contours→8,PlotPionts→50,
                PlotRange→{-0.5,0.5},ContourShading→False]
```

输入 In[4]命令，执行后得到 30 条等高线并去掉阴影的图形（如图 5-3）；输入 In[5]命令执行后得到 -0.5 与 0.5 之间的 30 条等高线（如图 5-4）：

```
In[4]:= ContourPlot[f[x, y], {x, -1, 1}, {y, -1, 1},
          Contours → 30, ContourShading → False]
```

```
In[5]:= ContourPlot[f[x, y], {x, -1, 1}, {y, -1, 1}, Contours → 30,
         PlotRange → {-0.5, 0.5}, ContourShading → False]
```



图 5-3



图 5-4



图 5-5

我们还可以画出自己规定的等高线，例如执行命令：

```
In[6]:= cont = {-0.8, -0.6, -0.4, -0.2, 0.2, 0.4, 0.6, 0.8};
        ContourPlot[f[x, y], {x, -1, 1}, {y, -1, 1},
         Contours → cont, ColorFunction → Hue]
```

得到了8条等高线，这些等高线分别是：

$$\frac{2x^3}{x^2+y^2}=C\ (C=-0.8,-0.6,-0.4,-0.2,0.2,0.4,0.6,0.8),$$

其中选项**ColorFunction→Hue**是指用一系列的色彩来表示等高线的明暗度。

2、 二元函数的梯度

设函数 $z=f(x,y)$ 在平面区域 D 内有一阶连续偏导数，则此函数在点 $P(x,y)$ 处的梯度为 $gradf(x,y)=\frac{\partial f}{\partial x}\vec{i}+\frac{\partial f}{\partial y}\vec{j}$，且我们知道沿梯度方向的方向导数达到最大值，梯度方向是函数 $z=f(x,y)$ 在这点增长最快的方向，及梯度方向与等高线的法线方向相同。

设 L 为平面曲线，如果 L 上任一点处的切线与函数 $z=f(x,y)$ 在该点处的梯度位于同

一直线上，则称 L 为 $z=f(x,y)$ 的梯度线。下面来讨论如何作出函数 $z=f(x,y)$ 的梯度线。

可以以等长的折线来模拟函数的梯度线。设步长为 λ，从点 $P_0(x_0,y_0)$ 出发，沿梯度方向前进 λ 得到点 $P_1(x_1,y_1)$，其中 $\begin{cases} x_1 = x_0 + \dfrac{f_x(x_0,y_0)}{\sqrt{f_x^{\ 2}(x_0,y_0)+f_y^{\ 2}(x_0,y_0)}}\cdot\lambda \\ y_1 = y_0 + \dfrac{f_y(x_0,y_0)}{\sqrt{f_x^{\ 2}(x_0,y_0)+f_y^{\ 2}(x_0,y_0)}}\cdot\lambda \end{cases}$，再从点 $P_1(x_1,y_1)$ 出发，沿梯度方向前进 λ 得到点 $P_2(x_2,y_2)$，这样依次得到一列点，连接这些点即可得到梯度线的图形。

例 2 作出函数 $f(x,y)=xe^y$ 的等高线和梯度线的图形。

解：（1）定义函数及两个偏导函数，并绘出 $z=f(x,y)$ 的图形。输入以下语句：

```
In[8]:= g[x_, y_] := x Exp[y];
        gx[x_, y_] := Evaluate[D[g[x, y], x]];
        gy[x_, y_] := Evaluate[D[g[x, y], y]];
        Plot3D[g[x, y], {x, -2, 2}, {y, -2, 2}]
```

运行结果如图 5-6。

（2）绘制等高线。执行下面命令即可得图 5-7 所示的等高线。

```
In[10]:= t1 = ContourPlot[g[x, y], {x, -2, 2}, {y, -2, 2},
           Contours → 10, ContourShading → False]
```



图 5-6



图 5-7

（3）绘制梯度线（以原点(0,0)为起始点）。输入如下语句：

```
In[11]:= x0 = 0; y0 = 0; lamda = 0.01; a = x0; b = y0;
         Do[
          u =
           a + lamda * gx[a, b] / Sqrt[(gx[a, b])^2 + (gy[a, b])^2];
          v =
           b + lamda * gy[a, b] / Sqrt[(gx[a, b])^2 + (gy[a, b])^2];
          c[n] = u; d[n] = v; a = u; b = v, {n, 300}];
         data = Table[{c[n], d[n]}, {n, 300}];
         t2 = ListPlot[data, Joined → True,
           PlotStyle → RGBColor[1, 0, 0]]
```

为了让梯度线和等高线相区别，这里限定了梯度线显示为红色。运行后得到如图 5-8 所示的梯度线。

（4）在同一坐标下显示等高线和梯度线。输入命令

```
In[14]:= Show[t1, t2, AspectRatio → Automatic, PlotRange → All]
```

运行后即得图 5-9。



图5-8



图5-9

3、 隐函数图形的绘制

在前面的实验中介绍了一元显函数及参数方程的作图函数，下面来讨论一元隐函数的作图问题。设由 $f(x,y)=C$ 确定了一个隐函数，它的图形即为二元函数 $z=f(x,y)$ 取值为 C 的那条等高线，因此我们可以通过等高线来作出一元隐函数的图形。在作图的方法很多，可以对二元函数的因变量要选取适当的值、等高线的条数取为1、去掉等高线中的阴影来作图；也可以利用选项**Contours→{C}**来作图；也可以直接用下面命令来作图：

ContourPlot[f[x,y]==C,{x，xmin,xmax},{y,ymin,ymax},选项]

例如执行以下三个命令后均可以得到圆 $x^2+y^2=4$ 的图形（如图5-10所示）：

```
In[15]:= ContourPlot[x^2 + y^2, {x, -2, 2}, {y, -2, 2},
          Contours → 1, ContourShading → False]
         ContourPlot[x^2 + y^2, {x, -3, 3}, {y, -3, 3},
          Contours → {4}, ContourShading → False]
         ContourPlot[x^2 + y^2 == 4, {x, -3, 3}, {y, -3, 3}]
```



图 5-10

例3 绘制隐函数 $x^2+y^2+e^{\frac{x^2}{2}}-20=0,\ -3\le x\le 3,-5\le y\le 5$ 的图形。

解：设 $f(x,y)=x^2+y^2+e^{\frac{x^2}{2}}$，则问题化为绘制二元函数的高度值为20的那条等高线。输入两个命令，执行后均可得到所求隐函数的图形（如图5-11所示）。

```
In[18]:= ContourPlot[x^2 * y^2 + Exp[x^2 / 2], {x, -3, 3},
          {y, -5, 5}, Contours → {20}, ContourShading → False]
         ContourPlot[x^2 * y^2 + Exp[x^2 / 2] == 20, {x, -3, 3},
          {y, -5, 5}]
```



图 5-11

类似地，可以利用命令**ContourPlot3D**来绘制二元隐函数的图形，其用法与命令**ContourPlot**相同。例如，执行以下两个命令，均可以得到球面 $x^2+y^2+z^2=4$ 的图形（如图5-12所示）。

```
In[20]:= ContourPlot3D[x^2 + y^2 + z^2 == 4, {x, -2, 2},
          {y, -2, 2}, {z, -2, 2}]
         ContourPlot3D[x^2 + y^2 + z^2, {x, -2, 2}, {y, -2, 2},
          {z, -2, 2}, Contours → {4}]
```



图5-12



图5-13

而下面命令的执行结果如图5-13所示：

```
In[22]:= ContourPlot3D[x^2 + y^2 + z^2, {x, -2, 2}, {y, -2, 2},
          {z, -2, 2}]
```

Mathematica智能地将其转化为隐函数 $x^2+y^2+z^2=a$，并绘制了多个 a 的可能取值的图形。

实验习题五

1、 作出函数 $f(x,y)=e^{-(x^2+2y^2)/10^4}$ 的等高线和梯度线的图形。

2、 作出下列隐函数的曲线：

（1） $x^4+y^4=16$　　　　（2） $xy=e^{x+y}$